NAG Fortran Library Routine Document

G08AAF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G08AAF performs the Sign test on two related samples of size n.

2 Specification

```
SUBROUTINE G08AAF(X, Y, N, IS, N1, P, IFAIL)
INTEGER           N, IS, N1, IFAIL
real              X(N), Y(N), P
```

3 Description

The Sign test investigates the median difference between pairs of scores from two matched samples of size n, denoted by $\{x_i, y_i\}$, for $i = 1, 2, \ldots, n$. The hypothesis under test, H_0, often called the null hypothesis, is that the medians are the same, and this is to be tested against a one- or two-sided alternative H_1 (see below).

G08AAF computes:

(a) the test statistic S, which is the number of pairs for which $x_i < y_i$;

(b) the number n_1 of non-tied pairs $(x_i \neq y_i)$;

(c) the lower tail probability p corresponding to S (adjusted to allow the complement $(1 - p)$ to be used in an upper one-tailed or a two-tailed test). p is the probability of observing a value $\leq S$ if $S < \frac{1}{2}n_1$, or of observing a value $< S$ if $S > \frac{1}{2}n_1$, given that H_0 is true. If $S = \frac{1}{2}n_1$, p is set to 0.5.

Suppose that a significance test of a chosen size α is to be performed (i.e., α is the probability of rejecting H_0 when H_0 is true; typically α is a small quantity such as 0.05 or 0.01). The returned value of p can be used to perform a significance test on the median difference, against various alternative hypotheses H_1, as follows

(i) H_1: median of $x \neq$ median of y. H_0 is rejected if $2 \times \min(p, 1 - p) < \alpha$.

(ii) H_1: median of $x >$ median of y. H_0 is rejected if $p < \alpha$.

(iii) H_1: median of $x <$ median of y. H_0 is rejected if $1 - p < \alpha$.

4 References

Siegel S (1956) *Nonparametric Statistics for the Behavioral Sciences* McGraw-Hill

5 Parameters

1: X(N) – ***real*** array *Input*

2: Y(N) – ***real*** array *Input*

On entry: X(i) and Y(i) must be set to the ith pair of data values, $\{x_i, y_i\}$, for $i = 1, 2, \ldots, n$.

3: N – INTEGER *Input*

On entry: the size of each sample, n.

Constraint: N ≥ 1.

4: IS – INTEGER *Output*

On exit: the Sign test statistic, S.

5: N1 – INTEGER *Output*

On exit: the number of non-tied pairs, n_1.

6: P – ***real*** *Output*

On exit: the lower tail probability, p, corresponding to S.

7: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. Users who are unfamiliar with this parameter should refer to Chapter P01 for details.

On exit: IFAIL $= 0$ unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, for users not familiar with this parameter the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL $= 1$

On entry, N < 1.

IFAIL $= 2$

N1 $= 0$, i.e., the samples are identical.

7 Accuracy

The tail probability, p, is computed using the relationship between the binomial and beta distributions. For $n_1 < 120$, p should be accurate to at least 4 significant figures, assuming that the machine has a precision of 7 or more digits. For $n_1 \geq 120$, p should be computed with an absolute error of less than 0.005. For further details see G01EEF.

8 Further Comments

The time taken by the routine is small, and increases with n.

9 Example

This example is taken from page 69 of Siegel (1956). The data relates to ratings of 'insight into paternal discipline' for 17 sets of parents, recorded on a scale from 1 to 5.

9.1 Program Text

Note: the listing of the example program presented below uses ***bold italicised*** terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     G08AAF Example Program Text
*     Mark 14 Revised.  NAG Copyright 1989.
*     .. Parameters ..
      INTEGER          N
      PARAMETER        (N=17)
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
*     .. Local Scalars ..
      real             SIG
      INTEGER          IFAIL, IS, N1
*     .. Local Arrays ..
      real             X(N), Y(N)
*     .. External Subroutines ..
      EXTERNAL         G08AAF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'G08AAF Example Program Results'
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) X, Y
      WRITE (NOUT,*)
      WRITE (NOUT,*) 'Sign test'
      WRITE (NOUT,*)
      WRITE (NOUT,*) 'Data values'
      WRITE (NOUT,*)
      WRITE (NOUT,99999) X, Y
      IFAIL = 0
*
      CALL G08AAF(X,Y,N,IS,N1,SIG,IFAIL)
*
      WRITE (NOUT,*)
      WRITE (NOUT,99998) 'Test statistic   ', IS
      WRITE (NOUT,99998) 'Observations     ', N1
      WRITE (NOUT,99997) 'Lower tail prob. ', SIG
      STOP
*
99999 FORMAT (4X,17F3.0)
99998 FORMAT (1X,A,I5)
99997 FORMAT (1X,A,F5.3)
      END
```

9.2 Program Data

```
G08AAF Example Program Data
 4 4 5 5 3 2 5 3 1 5 5 5 4 5 5 5 5
 2 3 3 3 3 3 3 3 2 3 2 2 5 2 5 3 1
```

9.3 Program Results

```
 G08AAF Example Program Results

 Sign test

 Data values

     4. 4. 5. 5. 3. 2. 5. 3. 1. 5. 5. 5. 4. 5. 5. 5. 5.
     2. 3. 3. 3. 3. 3. 3. 3. 2. 3. 2. 2. 5. 2. 5. 3. 1.

 Test statistic       3
 Observations        14
 Lower tail prob. 0.029
```